
25 Sequential Problems

This chapter extends simple games to a sequential context with multiple states.
A Markov game (MG) can be viewed as a Markov decision processe involving
multiple agents with their own reward functions.1 In this formulation, transitions

1 Markov games, also called
stochastic games, were originally
studied in the 1950s around the
same time as MDPs. L. S. Shapley,
‘‘Stochastic Games,’’ Proceedings
of the National Academy of Sciences,
vol. 39, no. 10, pp. 1095–1100, 1953.
They were introduced into the
multiagent artificial intelligence
community decades later. M. L.
Littman, ‘‘Markov Games as
a Framework for Multi-Agent
Reinforcement Learning,’’ in
International Conference on Machine
Learning (ICML), 1994.

depend on the joint action and all agents seek to maximize their own reward. We
generalize the response models and the Nash equilibrium solution concept from
simple games to take into account the state transition model. The last part of this
chapter discusses learning-based models where the agents adapt their policies
based on information from observed interactions and knowledge of the reward
and transition functions.

25.1 Markov Games

An MG (algorithm 25.1) is an extension of simple games to include a shared state
s ∈ S . The likelihood of transitioning from a state s to a state s′ under a joint
action a is given by the transition distribution T(s′ | s, a). Each agent i receives
a reward according to its own reward function Ri(s, a), which now additionally
depends on the state. Example 25.1 sketches how traffic routing can be framed as
a Markov game.

struct MG

γ # discount factor

ℐ # agents

𝒮 # state space

𝒜 # joint action space

T # transition function

R # joint reward function

end

Algorithm 25.1. Data structure for
a Markov game.

Consider commuters headed to work by car. Each car has a starting position
and a destination. Each car can take any of several available roads to get to
their destination, but these roads vary in the time it takes to drive them. The
more cars that drive on a given road, the slower they all move.

This problem is a Markov game. The agents are the commuters in their
cars, the states are the locations of all the cars on the roads, and the actions
correspond to decisions onwhich road to take next. The state transitionmoves
all car agents forward following their joint action. The negative reward is
proportional to the time spent driving on a road.

Example 25.1. Traffic routing as a
Markov game. The problem cannot
be modeled using a single agent
model like an MDP because we
do not know the behavior of other
agents, only their rewards. We can
try to find equilibria or learn poli-
cies through interaction, similar to
what we did for simple games.

The joint policy π in a Markov game specifies a probability distribution over
joint actions given the current state. As with Markov decision processes, we will
focus on policies that depend on the current state rather than the past history
because future states and rewards are conditionally independent of the history
given the current state. In addition, we will focus on stationary policies, which
do not depend on time. The probability agent i selects action a at state s is given
by πi(a | s). We will often use π(s) to represent a distribution over joint actions.

The utility of a joint policy π from the perspective of agent i can be computed
using a variation of policy evaluation introduced in section 7.2 for MDPs. The
reward to agent i from state s when following joint policy π is

Ri(s,π(s)) = ∑
a

Ri(s, a)∏
j∈I

π j(aj | s) (25.1)

The probability of transitioning from state s to s′ when following π is

T(s′ | s,π(s)) = ∑
a

T(s′ | s, a)∏
j∈I

π j(aj | s) (25.2)

In an infinite horizon, discounted game, the utility for agent i from state s is

Uπ,i(s) = Ri(s,π(s)) + γ ∑
s′

T(s′ | s,π(s))Uπ,i(s′) (25.3)

which can be solved exactly (algorithm 25.2).

struct MGPolicy

p # dictionary mapping states to simple game policies

MGPolicy(p::Base.Generator) = new(Dict(p))

end

(πi::MGPolicy)(s, ai) = πi.p[s](ai)

(πi::SimpleGamePolicy)(s, ai) = πi(ai)

probability(𝒫::MG, s, π, a) = prod(πj(s, aj) for (πj, aj) in zip(π, a))

reward(𝒫::MG, s, π, i) =

sum(𝒫.R(s,a)[i]*probability(𝒫,s,π,a) for a in joint(𝒫.𝒜))

transition(𝒫::MG, s, π, s′) =

sum(𝒫.T(s,a,s′)*probability(𝒫,s,π,a) for a in joint(𝒫.𝒜))

function policy_evaluation(𝒫::MG, π, i)

𝒮, 𝒜, R, T, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T, 𝒫.γ

p(s,a) = prod(πj(s, aj) for (πj, aj) in zip(π, a))

R′ = [sum(R(s,a)[i]*p(s,a) for a in joint(𝒜)) for s in 𝒮]

T′ = [sum(T(s,a,s′)*p(s,a) for a in joint(𝒜)) for s in 𝒮, s′ in 𝒮]

return (I - γ*T′)\R′

end

Algorithm 25.2. A Markov game
policy is a mapping from states to
simple game policies, introduced
in the previous chapter. We can
construct it by passing in a genera-
tor to construct the dictionary. The
probability a policy (either for a
Markov game or simple game) as-
signs to taking action ai from state
s is πi(s,ai). Functions are also
provided for computing Ri(s,π(s))
and T(s′ | s,π(s)). The policy eval-
uation function will compute a vec-
tor representing Uπ,i .

25.2 Response Models

We can generalize the response models introduced in the previous chapter to
Markov games. Doing so requires taking into account the state transition model.

25.2.1 Best Response

A response policy for agent i is a policy πi that maximizes expected utility given
the fixed policies of other agents π−i. If the policies of the other agents are fixed,
then the problem reduces to an MDP. This MDP has state space S and action
space Ai. We can define the transition and reward functions as follows:

T′(s′ | s, ai) = T(s′ | s, ai,π−i(s)) (25.4)
R′(s, ai) = Ri(s, ai,π−i(s)) (25.5)

Because this is a best response for agent i, the MDP only uses reward Ri. Solving
this MDP results in a best response policy for agent i. Algorithm 25.3 provides an
implementation.

512 chapter 25. sequential problems

function best_response(𝒫::MG, π, i)

𝒮, 𝒜, R, T, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T, 𝒫.γ

T′(s,ai,s′) = transition(𝒫, s, joint(π, SimpleGamePolicy(ai), i), s′)

R′(s,ai) = reward(𝒫, s, joint(π, SimpleGamePolicy(ai), i), i)

πi = solve(MDP(γ, 𝒮, 𝒜[i], T′, R′))

return MGPolicy(s => SimpleGamePolicy(πi(s)) for s in 𝒮)

end

Algorithm 25.3. For a Markov
game 𝒫, we can compute a deter-
ministic best response policy for
agent i given that the other agents
are playing policies in π. We can
solve the MDP exactly using one of
the methods from chapter 7.

25.2.2 Softmax Response

Similar to what was done in the previous chapter, we can define a softmax response
policy, which assigns a stochastic response to the policies of the other agents at
each state. As we did in the construction of a deterministic best response policy,
we solve an MDP where the agents with the fixed policies π−i are folded into the
environment. We then extract the action value function Q(s, a) using one-step
lookahead. The softmax response is then

πi(ai | s) ∝ exp(λQ(s, ai)) (25.6)

with precision parameter λ ≥ 0. Algorithm 25.4 provides an implementation. This
approach can be used to generate hierarchical softmax solutions (section 24.6).
In fact, we can use algorithm 24.9 directly.

function softmax_response(𝒫::MG, π, i, λ)

𝒮, 𝒜, R, T, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T, 𝒫.γ

T′(s,ai,s′) = transition(𝒫, s, joint(π, SimpleGamePolicy(ai), i), s′)

R′(s,ai) = reward(𝒫, s, joint(π, SimpleGamePolicy(ai), i), i)

mdp = MDP(γ, 𝒮, joint(𝒜), T′, R′)

πi = solve(mdp)

Q(s,a) = lookahead(mdp, πi.U, s, a)

p(s) = SimpleGamePolicy(a => exp(λ*Q(s,a)) for a in 𝒜[i])

return MGPolicy(s => p(s) for s in 𝒮)

end

Algorithm 25.4. The softmax re-
sponse of agent i to joint policy π

with precision parameter λ.

25.3 Nash Equilibrium

The Nash equilibrium concept can be generalized to Markov games.2 As with

2 Because we assume policies are
stationary, in that they do not vary
over time, the Nash equilibria cov-
ered here are stationary Markov per-
fect equilibria.

simple games, all agents perform a best response to one another, and have no
incentive to deviate. All finite Markov games with a discounted infinite horizon
have a Nash equilibrium.3

3 A.M. Fink, ‘‘Equilibrium in a
Stochastic n-Person Game,’’ Journal
of Science of the Hiroshima University,
Series A-I, vol. 28, no. 1, pp. 89–93,
1964.

We can find a Nash equilibrium by solving a nonlinear optimization problem
similar to the one we solved in the context of simple games. This problem mini-
mizes the sum of the lookahead utility deviations and constrains the policies to
be valid distributions:

minimize
π,U

∑
i∈I

∑
s

(
Ui(s)−Qi(s,π(s))

)
subject to Ui(s) ≥ Qi(s, ai,π−i(s)) for all i, s, ai

∑
ai

πi(ai | s) = 1 for all i, s

πi(ai | s) ≥ 0 for all i, s, ai

(25.7)

where
Qi(s,π(s)) = Ri(s,π(s)) + γ ∑

s′
T(s′ | s,π(s))Ui(s′) (25.8)

This nonlinear optimization problem is implemented and solved in algorithm25.5.4 4 J. A. Filar, T.A. Schultz, F. Thui-
jsman, and O. Vrieze, ‘‘Nonlin-
ear Programming and Stationary
Equilibria in Stochastic Games,’’
Mathematical Programming, vol. 50,
no. 1–3, pp. 227–237, 1991.

25.4 Opponent Modeling

Aswedid in the context of simple games,we can take a learning-based approach to
arrive at joint policies by running agents in simulation. Algorithm 25.6 generalizes
the simulation loop introduced in the previous chapter to handle state transitions.
The various policies run in simulation update themselves based on the state
transitions and the actions taken by the various agents.

One approach for updating policies is opponent modeling (algorithm 25.7).5 Sim- 5 This algorithm was conceived
by Uther and Veloso and Bowl-
ing and Veloso as an extension of
fictitious play to Markov games.
W. Uther and M. Veloso, ‘‘Adver-
sarial Reinforcement Learning,’’
Carnegie Mellon University, Tech.
Rep. CMU-CS-03-107, 1997. M.
Bowling and M. Veloso, ‘‘An Anal-
ysis of Stochastic Game Theory for
Multiagent Reinforcement Learn-
ing,’’ Carnegie Mellon University,
Tech. Rep. CMU-CS-00-165, 2000.

ilar to fictitious play (section 24.7), it involves maintaining a maximum-likelihood
model over the policies of the other agents. However, opponent modeling takes
advantage of a known transition function.

The maximum likelihood models track the state in addition to the action being
taken by each agent. We track the number of times agent j takes action aj in state s,
storing it in a table N(j, aj, s), typically initialized to 1. Then, we can compute the
best response assuming that each agent j follows the state-dependent stochastic
policy:

π j(aj | s) ∝ Ni(j, aj, s) (25.9)

After observing joint action a in states s, we update

N(j, aj, s)← N(j, aj, s) + 1 (25.10)

function tensorform(𝒫::MG)

ℐ, 𝒮, 𝒜, R, T = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜, 𝒫.R, 𝒫.T

ℐ′ = eachindex(ℐ)

𝒮′ = eachindex(𝒮)

𝒜′ = [eachindex(𝒜[i]) for i in ℐ]

R′ = [R(s,a) for s in 𝒮, a in joint(𝒜)]

T′ = [T(s,a,s′) for s in 𝒮, a in joint(𝒜), s′ in 𝒮]

return ℐ′, 𝒮′, 𝒜′, R′, T′

end

function solve!(M::NashEquilibrium, 𝒫::MG)

ℐ, 𝒮, 𝒜, R, T = tensorform(𝒫)

𝒮′, 𝒜′, γ = 𝒫.𝒮, 𝒫.𝒜, 𝒫.γ

model = Model(Ipopt.Optimizer)

@variable(model, U[ℐ, 𝒮])

@variable(model, π[i=ℐ, 𝒮, ai=𝒜[i]] ≥ 0)

@NLobjective(model, Min,

sum(U[i,s] - sum(prod(π[j,s,a[j]] for j in ℐ)

* (R[s,y][i] + γ*sum(T[s,y,s′]*U[i,s′] for s′ in 𝒮))

for (y,a) in enumerate(joint(𝒜))) for i in ℐ, s in 𝒮))

@NLconstraint(model, [i=ℐ, s=𝒮, ai=𝒜[i]],

U[i,s] ≥ sum(

prod(j==i ? (a[j]==ai ? 1.0 : 0.0) : π[j,s,a[j]] for j in ℐ)

* (R[s,y][i] + γ*sum(T[s,y,s′]*U[i,s′] for s′ in 𝒮))

for (y,a) in enumerate(joint(𝒜))))

@constraint(model, [i=ℐ, s=𝒮], sum(π[i,s,ai] for ai in 𝒜[i]) == 1)

optimize!(model)

π′ = value.(π)

πi′(i,s) = SimpleGamePolicy(𝒜′[i][ai] => π′[i,s,ai] for ai in 𝒜[i])

πi′(i) = MGPolicy(𝒮′[s] => πi′(i,s) for s in 𝒮)

return [πi′(i) for i in ℐ]

end

Algorithm 25.5. This nonlinear
program computes a Nash equilib-
rium for a Markov game 𝒫.

function randstep(𝒫::MG, s, a)

s′ = rand(SetCategorical(𝒫.𝒮, [𝒫.T(s, a, s′) for s′ in 𝒫.𝒮]))

r = 𝒫.R(s,a)

return s′, r

end

function simulate(𝒫::MG, π, k_max, b)

s = rand(b)

for k = 1:k_max

a = Tuple(πi(s)() for πi in π)

s′, r = randstep(𝒫, s, a)

for πi in π

update!(πi, s, a, s′)

end

s = s′

end

return π

end

Algorithm 25.6. Functions for tak-
ing a random step and running full
simulations in Markov games. The
simulate function will simulate the
joint policy π for k_max steps start-
ing from a state randomly sampled
from b.

for each agent j.
As the distributions of the other agents’ actions change, we must update the

utilities. The utilities in opponent modeling are significantly more difficult to
compute. As described in section 25.2.1, for any assignment of fixed policies of
others π−i determined from equation (25.9) induces an MDP. Instead of solving
an MDP at each update, it is common to apply the update periodically, a strategy
adopted from asynchronous value iteration. An example of opponent modeling
is given in example 25.2.

Our policy πi(s) for a state s is derived from a given opponent model π−i and
computed utility Ui. We then select a best response:

arg max
a

Qi(s, a,π−i) (25.11)

In the implementation here, we use the property that each state of a Markov game
policy is a simple game policy whose reward is the corresponding Qi.

25.5 Nash Q-Learning

Another learning-based approach is Nash Q-learning (algorithm 25.8), which
borrows inspiration from Q-learning (section 17.2).6 The method maintains an

6 J. Hu and M.P. Wellman, ‘‘Nash
Q-Learning for General-Sum
Stochastic Games,’’ Journal of
Machine Learning Research, vol. 4,
no. November, pp. 1039–1069,
2003.estimate of the action value function, which is adapted as the agents react to each

mutable struct OpponentModeling

𝒫 # Markov game

i # agent index

Qi # state-action value estimates

Ni # state-action counts

end

function OpponentModeling(𝒫::MG, i)

ℐ, 𝒮, 𝒜, R = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜, 𝒫.R

Qi = Dict((s, a) => R(s, a)[i] for s in 𝒮 for a in joint(𝒜))

Ni = Dict((j, s, aj) => 1.0 for j in ℐ for s in 𝒮 for aj in 𝒜[j])

return OpponentModeling(𝒫, i, Qi, Ni)

end

function (πi::OpponentModeling)(s)

𝒫, i, Qi = πi.𝒫, πi.i, πi.Qi

ℐ, 𝒮, 𝒜, T, R, γ = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜, 𝒫.T, 𝒫.R, 𝒫.γ

πi′(i,s) = SimpleGamePolicy(ai => πi.Ni[i,s,ai] for ai in 𝒜[i])

πi′(i) = MGPolicy(s => πi′(i,s) for s in 𝒮)

π = [πi′(i) for i in ℐ]

U(s,π) = sum(πi.Qi[s,a]*probability(𝒫,s,π,a) for a in joint(𝒜))

Q(s,π) = reward(𝒫,s,π,i) + γ*sum(transition(𝒫,s,π,s′)*U(s′,π)

for s′ in 𝒮)

Q(ai) = Q(s, joint(π, SimpleGamePolicy(ai), i))

ai = _argmax(Q, 𝒫.𝒜[πi.i])

return SimpleGamePolicy(ai)

end

function update!(πi::OpponentModeling, s, a, s′)

𝒫, i, Qi = πi.𝒫, πi.i, πi.Qi

ℐ, 𝒮, 𝒜, T, R, γ = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜, 𝒫.T, 𝒫.R, 𝒫.γ

for (j,aj) in enumerate(a)

πi.Ni[j,s,aj] += 1

end

πi′(i,s) = SimpleGamePolicy(ai => πi.Ni[i,s,ai] for ai in 𝒜[i])

πi′(i) = MGPolicy(s => πi′(i,s) for s in 𝒮)

π = [πi′(i) for i in ℐ]

U(π,s) = sum(πi.Qi[s,a]*probability(𝒫,s,π,a) for a in joint(𝒜))

Q(s,a) = R(s,a)[i] + γ*sum(T(s,a,s′)*U(π,s′) for s′ in 𝒮)

for a in joint(𝒜)

πi.Qi[s,a] = Q(s,a)

end

end

Algorithm 25.7. Opponent mod-
eling for agent i in an MG 𝒫 that
maintains counts Ni of other agent
action selections over time for each
state and averages them assum-
ing this is their stochastic policy. It
then computes a best response to
this policy and performs the cor-
responding utility-maximizing ac-
tion.

The predator-prey hex worldMarkov game (appendix F.13) has one predator
(red) and one prey (blue). If the predator catches the prey, it receives a reward
of 10 and the prey receives a reward of −100. Otherwise, both agents receive
a−1 reward. The agents move simultaneously. We apply opponent modeling
with resets to the initial state every 10 steps.

We observe that the predator learns to chase the prey and the prey learns
to flee. Interestingly, the predator also learns that the prey runs to the right
corner and waits. Here, the prey learns that if it waits at this corner, it can
flee from the predator immediately as it jumps towards the prey. In this case,
it escapes the predator and runs to the other side of the map.

Below is a plot of the learned opponent model of the highlighted state
(both predator and prey hex locations) for both the predator and the prey.

Example 25.2. An example of op-
ponent modeling on the predator-
prey hex world problem.

other’s changing policies. In the process of updating the action value function, it
computes a Nash equilibrium to model the behavior of the other agents.

An agent following Nash Q-learning maintains an estimate of a joint action
value function Q(s, a). This action value function is updated after every state
transition using a Nash equilibrium computed from a simple game constructed
from this value function. After a transition from s to s′ following the joint action
a, we construct a simple game with the same number of agents and the same joint
action space, but have the reward function equal to the estimated value of s′ such
that R(a′) = Q(s′, a′). The agent computes a Nash equilibrium policy π′ over the
next action a′. Under the derived policy, the expected utility of the successor state
is:

U(s′) = ∑
a′

Q(s′, a′)∏
j∈I

π j ′(aj ′) (25.12)

The agent then updates its value function:

Q(s, a)← Q(s, a) + α
(
R(s, a) + γU(s′)−Q(s, a)

)
(25.13)

where the learning rate α is typically a function of the state-action count α =

1/
√

N(s, a).
As with regular Q-learning, we need to adopt an exploration strategy to ensure

that all states and actions are tried sufficiently often. In algorithm 25.8, the agent
follows an ε-greedy policy. With probability ε = 1/ ∑a(N(s, a)), it selects an
action uniformly at random. Otherwise, it will use the result from the Nash
equilibrium.

25.6 Summary

• Markov games are an extension of MDPs to multiple agents, or an extension
of simple games to sequential problems. In these problems, multiple agents
compete and individually receive reward over time.

• The Nash equilibrium can be formulated for Markov games, but must now
consider all actions for all agents in all states.

• The problem of finding a Nash equilibrium can be formulated as a nonlinear
optimization problem.

mutable struct NashQLearning

𝒫 # Markov game

i # agent index

Q # state-action value estimates

N # history of actions performed

end

function NashQLearning(𝒫::MG, i)

ℐ, 𝒮, 𝒜 = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜

Q = Dict((j, s, a) => 0.0 for j in ℐ, s in 𝒮, a in joint(𝒜))

N = Dict((s, a) => 1.0 for s in 𝒮, a in joint(𝒜))

return NashQLearning(𝒫, i, Q, N)

end

function (πi::NashQLearning)(s)

𝒫, i, Q, N = πi.𝒫, πi.i, πi.Q, πi.N

ℐ, 𝒮, 𝒜, 𝒜i, γ = 𝒫.ℐ, 𝒫.𝒮, 𝒫.𝒜, 𝒫.𝒜[πi.i], 𝒫.γ

M = NashEquilibrium()

𝒢 = SimpleGame(γ, ℐ, 𝒜, a -> [Q[j, s, a] for j in ℐ])

π = solve(M, 𝒢)

ϵ = 1 / sum(N[s, a] for a in joint(𝒜))

πi′(ai) = ϵ/length(𝒜i) + (1-ϵ)*π[i](ai)

return SimpleGamePolicy(ai => πi′(ai) for ai in 𝒜i)

end

function update!(πi::NashQLearning, s, a, s′)

𝒫, ℐ, 𝒮, 𝒜, R, γ = πi.𝒫, πi.𝒫.ℐ, πi.𝒫.𝒮, πi.𝒫.𝒜, πi.𝒫.R, πi.𝒫.γ

i, Q, N = πi.i, πi.Q, πi.N

M = NashEquilibrium()

𝒢 = SimpleGame(γ, ℐ, 𝒜, a′ -> [Q[j, s′, a′] for j in ℐ])

π = solve(M, 𝒢)

πi.N[s, a] += 1

α = 1 / sqrt(N[s, a])

for j in ℐ

πi.Q[j,s,a] += α*(R(s,a)[j] + γ*utility(𝒢,π,j) - Q[j,s,a])

end

end

Algorithm 25.8. Nash Q-learning
for an agent i in an MG 𝒫. The al-
gorithm performs joint-action Q-
learning to learn a state-action
value function for all agents. A sim-
ple game is built with Q, and we
compute a Nash equilibrium using
algorithm 24.5. The equilibrium
is then used to update the value
function. This implementation also
uses a variable learning rate pro-
portional to the number of times
state-joint-action pairs are visited,
which is stored in N. Additionally,
it uses ε-greedy exploration to en-
sure all states and actions are ex-
plored.

520 chapter 25. sequential problems

• Opponent modeling is a learning-based approach similar to fictitious play, but
it involves using a known transition function.

• Nash Q-learning adapts traditional Q-learning to multiagent problems and
involves solving for a Nash equilibrium of a simple game constructed from
models of the other players.

25.7 Exercises

Exercise 25.1. Show how Markov games are extensions of both MDPs and simple games.
Show this by formulating an MDP as a Markov game and by formulating a simple game
as a Markov game.

Solution: Markov games generalize simple games. For any simple game with I , A, R, we
can construct a Markov game by just having a single state that self-loops. In other words,
this Markov game has S = {s1}, T(s1 | s1, a) = 1, and R(s1, a) = R(a).

Markov games generalize MDPs. For any MDP with S , A, T, and R, we can construct
a Markov game by just assigning the agents to be this single agent. In other words, this
Markov game has I = {1}, A1 = A, T(s′ | s, a) = T(s′ | s′, a), and R(s, a) = R(s, a).

Exercise 25.2. For an agent i, given the fixed policies of other agents π−i, can there exist a
stochastic best response that yields a greater utility than a deterministic best response?
Why then do we consider stochastic policies in a Nash equilibrium?

Solution: No, if given fixed policies of other agents π−i, a deterministic best response is
sufficient to obtain the highest utility. The best response can be formulated as solving an
MDP as described in section 25.2. It has been shown that deterministic policies are sufficient
to provide optimal utility-maximization. Hence, the same is true for a best response in a
Markov game.

In a Nash equilibrium, a best response has to hold for all agents. Although a determin-
istic best response might be equal in utility to a stochastic one, an equilibrium may require
stochastic responses to prevent other agents from wanting to deviate.

Exercise 25.3. This chapter discussed only stationary Markov policies. What other cate-
gories of policies are there?

Solution: A so-called behavioral policy πi(ht) is one that has a dependence on complete
history ht = (s1:t, a1:t−1). Such policies depend on the history of play of other agents. A
non-stationary Markov policy πi(s, t) is one that depends on the time step t, but not on the
complete history. For example, in the predator prey hex world domain, for the first 10 time
steps, the action at a hex might be to go east and after 10 time steps to go west.

There can exist Nash equlibria that are in the space of non-stationary non-Markov joint
policies, stationary non-Markov joint policies, and so forth. However, it is proven that
every (stationary) Markov game has a stationary Markov Nash equilibrium.

Exercise 25.4. Opponent modeling generalizes fictitious play in a simple game. In Markov
games, however, the utilities must also be computed, whereas in simple games this was not
necessary. List different approaches to compute utilities with their benefits and drawbacks.

Solution: The algorithm presented in this chapter performs a single backup for the visited
state s and all joint actions a. This approach has the benefit of being relatively efficient
because it is a single backup. Updating all joint actions at that state results in exploring
actions that were not observed. The drawback of this approach is that we may need to do
this update at all states many times to obtain a suitable policy.

An alternative is to only update the visited state and the joint action that was actually
taken, which results in a faster update step. The drawback is that it requires many more
steps to explore the full range of joint actions.

Another alternative is to perform value iteration at all states s until convergence at every
update step. Recall that the model of the opponent changes on each update. This induces a
new MDP, as described for deterministic best response in section 25.2.1. Consequently, we
would need to rerun value iteration after each update. The benefit of this approach is that
it can result in the most informed decision at each step because the utilities Qi consider all
states over time. The drawback is that the update step is very computationally expensive.

